HRP 262, SAS LAB THREE, April 29, 2009

Topics: Cox Regression

Lab Objectives

After today’s lab you should be able to:

1. Fit models using PROC PHREG.  Understand PROC PHREG output.

2. Understand how to implement and interpret different methods for dealing with ties (exact, efron, breslow, discrete).

3. Understand output from the “baseline” statement.

4. Output estimated survivor functions and plot cumulative hazards.  

5. Output and plot predicted survivor functions at user-specified levels of the covariates.

6. Understand the role of the strata statement in PROC PHREG.

7. Evaluate PH assumption graphically and by including interactions with time in the model.

8. Use the “where” subsetting statement in all PROC’s.

9. Use PROC FORMAT to format variables other than time.

10. Add time-dependent variables to the model.  Understand SAS syntax for time-dependent variables.  Be able to correctly specify the time-dependent variables you intend!

LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Go to the class website: www.stanford.edu/~kcobb/courses/hrp262 
Lab 3 data( Save(Save on your desktop as uis (SAS format)
(**if you don’t have hmohiv data in SAS format from last week)

Lab 2 data( Save(Save on your desktop as hmohiv)

2. Open SAS: Start Menu( All Programs(SAS
3. Name a library using the libname statement or point-and-click:
libname hrp262 'Extension to your desktop’;

4. We will first use the HMO-HIV dataset to examine ties.  Recall from last time that the HMOHIV dataset has many ties for time.
/**Do not specify ties**/

proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug / risklimits; 

title 'Cox model for hmohiv data-- ties';

run;

Cox model for hmohiv data-- ties                       

                                       The PHREG Procedure

                                        Model Information

                         Data Set                 HRP262.HMOHIV

                         Dependent Variable       Time

                         Censoring Variable       Censor          Censor

                         Censoring Value(s)       0

                         Ties Handling            BRESLOW

                       Summary of the Number of Event and Censored Values

                                                               Percent

                             Total       Event    Censored    Censored

                               100          80          20       20.00

                                       Convergence Status

                         Convergence criterion (GCONV=1E-8) satisfied.

                                      Model Fit Statistics

                                              Without           With

                             Criterion     Covariates     Covariates

                             -2 LOG L         598.390        563.408

                             AIC              598.390        567.408

                             SBC              598.390        572.172

                             Testing Global Null Hypothesis: BETA=0

                     Test                 Chi-Square       DF     Pr > ChiSq

                     Likelihood Ratio        34.9819        2         <.0001

                     Score                   34.3038        2         <.0001

                     Wald                    32.4859        2         <.0001


                            Analysis of Maximum Likelihood Estimates


                Parameter   Standard                         Hazard  95% Hazard Ratio Variable

   Variable DF   Estimate      Error Chi-Square Pr > ChiSq    Ratio Confidence Limits Label

   Age       1    0.09151    0.01849    24.5009     <.0001    1.096    1.057    1.136 Age

   Drug      1    0.94108    0.25550    13.5662     0.0002    2.563    1.553    4.229 Drug


6.  Rerun the Cox model with the other possible specifications for ties (use cut and paste to avoid retyping repeated code).  Then compare. 

/**Efron**/
proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug / ties=efron risklimits; 

title 'Cox model for hmohiv data—ties=efron';

run;

/**Exact**/
proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug / ties=exact risklimits; 

title 'Cox model for hmohiv data—ties=exact';

run;

/**Discrete**/
proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug / ties=discrete risklimits; 

title 'Cox model for hmohiv data—ties=discrete';

run;

RESULTS TO COMPARE:

                         Ties Handling            BRESLOW (from above)
                Parameter   Standard                         Hazard  95% Hazard Ratio Variable

   Variable DF   Estimate      Error Chi-Square Pr > ChiSq    Ratio Confidence Limits Label

   Age       1    0.09151    0.01849    24.5009     <.0001    1.096    1.057    1.136 Age

   Drug      1    0.94108    0.25550    13.5662     0.0002    2.563    1.553    4.229 Drug

                         Ties Handling            EFRON

                Parameter   Standard                         Hazard  95% Hazard Ratio Variable

   Variable DF   Estimate      Error Chi-Square Pr > ChiSq    Ratio Confidence Limits Label

  Age       1    0.09714    0.01864    27.1597     <.0001    1.102    1.062    1.143 Age

  Drug      1    1.01670    0.25622    15.7459     <.0001    2.764    1.673    4.567 Drug

                         Ties Handling            EXACT

                Parameter   Standard                         Hazard  95% Hazard Ratio Variable

   Variable DF   Estimate      Error Chi-Square Pr > ChiSq    Ratio Confidence Limits Label

   Age       1    0.09768    0.01874    27.1731     <.0001    1.103    1.063    1.144 Age

   Drug      1    1.02263    0.25716    15.8132     <.0001    2.781    1.680    4.603 Drug

                         Ties Handling            DISCRETE

                Parameter   Standard                         Hazard  95% Hazard Ratio Variable

   Variable DF   Estimate      Error Chi-Square Pr > ChiSq    Ratio Confidence Limits Label

   Age       1    0.10315    0.02006    26.4449     <.0001    1.109    1.066    1.153 Age

   Drug      1    1.07004    0.27438    15.2084     <.0001    2.916    1.703    4.992 Drug

7.  Estimate the survival function. SAS estimates the baseline hazard and survival functions using a nonparametric maximum likelihood method.

Using the estimated coefficients for the covariates (the betas from above), SAS can estimate a survival function for an individual with specific values of the covariates (SAS default plugs in mean values for the cohort).  Use a baseline statement to output this estimated survival function (and log survival and log(-log survival)) as follows. :

title ' ';
proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug / ties=discrete risklimits; 

baseline out=outdata survival=S logsurv=ls loglogs=lls;

run;

proc print data=outdata;

run; 


         
Obs     Age     Drug    Time       S          ls          lls


                   1    36.07    0.49      0     1.00000     0.00000      .

                   2    36.07    0.49      1     0.88817    -0.11859    -2.13208

                   3    36.07    0.49      2     0.84537    -0.16798    -1.78392

                   4    36.07    0.49      3     0.74452    -0.29501    -1.22073

                   5    36.07    0.49      4     0.69591    -0.36254    -1.01462

                   6    36.07    0.49      5     0.59867    -0.51304    -0.66740

                   7    36.07    0.49      6     0.56854    -0.56469    -0.57148

                   8    36.07    0.49      7     0.47163    -0.75156    -0.28560

                   9    36.07    0.49      8     0.39843    -0.92021    -0.08315

                  10    36.07    0.49      9     0.34557    -1.06256     0.06068

                  11    36.07    0.49     10     0.29593    -1.21762     0.19690

                  12    36.07    0.49     11     0.24697    -1.39848     0.33539

                  13    36.07    0.49     12     0.21396    -1.54196     0.43305

                  14    36.07    0.49     13     0.19278    -1.64620     0.49847

                  15    36.07    0.49     14     0.17243    -1.75774     0.56403

                  16    36.07    0.49     15     0.13433    -2.00743     0.69686

                  17    36.07    0.49     22     0.11498    -2.16296     0.77148

                  18    36.07    0.49     30     0.09637    -2.33952     0.84995

                  19    36.07    0.49     31     0.07970    -2.52948     0.92801

                  20    36.07    0.49     32     0.06467    -2.73842     1.00738

                  21    36.07    0.49     34     0.05111    -2.97371     1.08981

                  22    36.07    0.49     35     0.03913    -3.24085     1.17584

                  23    36.07    0.49     36     0.02767    -3.58750     1.27746

                  24    36.07    0.49     43     0.01766    -4.03670     1.39543

                  25    36.07    0.49     53     0.01038    -4.56754     1.51897

                  26    36.07    0.49     54     0.00554    -5.19588     1.64787

                  27    36.07    0.49     57     0.00176    -6.33964     1.84682

                  28    36.07    0.49     58     0.00028    -8.18864     2.10275


  8. Graph the cumulative hazard. Recall: cumulative hazard = -logS(t).


data outdata2;


set outdata;


ls=-ls;

run;

goptions reset=all;

axis1 label=(angle=90);

proc gplot data=outdata2;


label time='Time(Months)';


label ls='Cumulative Hazard';


plot ls*time / vaxis=axis1;


symbol1 value=none i=join;

run;

9. We can also use these plots to assess the validity of the PH assumption.  As discussed in lecture Monday, PH assumption implies that log-log survival curves should be parallel.



proc phreg data=hrp262.hmohiv;

model time*censor(0)= age/ ties=discrete risklimits; 

strata drug;

baseline out=outdata survival=S logsurv=ls loglogs=lls;

run;


proc gplot data=outdata;

title 'Evaluate proportional hazards assumption for variable: drug';

plot lls*time=drug /vaxis=axis1;

symbol1 i=join c=black line=1;

symbol2 i=join c=black line=2;

run;

10.  Since it’s hard to tell if the hazards are proportional, we might wonder if PH assumption is violated and if there is an interaction between drug and time.  Can test this (and fix problem) by adding such an interaction term to the model:
proc phreg data=hrp262.hmohiv;

model time*censor(0)=age drug drugtime/ ties=discrete risklimits; 

drugtime=drug*time;

label drugtime='Interaction of drug with time';

run;

RESULTS:

                       Analysis of Maximum Likelihood Estimates

                  Parameter    Standard                            Hazard   95% Hazard Ratio

   Variable  DF    Estimate       Error  Chi-Square  Pr > ChiSq     Ratio   Confidence Limits

   Age        1     0.10044     0.02036     24.3421      <.0001     1.106     1.062     1.151

   Drug       1     1.26833     0.37310     11.5564      0.0007     3.555     1.711     7.386

   drugtime   1    -0.03045     0.04060      0.5625      0.4533     0.970     0.896     1.050         


11. Repeat the same steps for age.  Note that age is continuous, but can easily be made categorical within the PHREG procedure!


proc phreg data=hrp262.hmohiv;

model time*censor(0)=/ ties=discrete risklimits; 

strata age(30,40);

baseline out=outdata2 survival=S logsurv=ls loglogs=lls;

run;

proc gplot data=outdata2;

title 'Evaluate proportional hazards assumption for variable: age';

plot lls*time=age /vaxis=axis1;

symbol1 i=join c=black line=1;

symbol2 i=join c=black line=2;

symbol3 i=join c=black line=3;

run;

proc phreg data=hrp262.hmohiv;

model time*censor(0)=agetime age/ ties=discrete risklimits; 

label agetime='Interaction of age with time';

agetime=age*time;

run;

RESULTS

                  Parameter    Standard                            Hazard   95% Hazard Ratio

   Variable  DF    Estimate       Error  Chi-Square  Pr > ChiSq     Ratio   Confidence Limits

   agetime    1  -0.0002915     0.00139      0.0439      0.8340     1.000     0.997     1.002

   Age        1     0.09470     0.02421     15.3058      <.0001     1.099     1.048     1.153

12. Obtain predictions about survival times for particular sets of covariate values that need not appear in the data set being analyzed. You have to create a new dataset that contains the covariate values of interest. 

data MyCovs;


input age drug;


datalines;


55
1


22
0


;

run;

proc phreg data=hrp262.hmohiv;


model time*censor(0) = drug age /ties=efron;


baseline out=outdata3 covariates=mycovs survival=S lower=LCL upper=UCL /nomean;

run;

proc print data=outdata3;

run;

     Obs    Drug    Age    Time       S         LCL        UCL


                     1      1      55      0     1.00000     .          .

                     2      1      55      1     0.27358    0.11377    0.65787

                     3      1      55      2     0.15991    0.04661    0.54861

                     4      1      55      3     0.04036    0.00522    0.31231

                     5      1      55      4     0.01954    0.00154    0.24731

                     6      1      55      5     0.00394    0.00011    0.14123

                     7      1      55      6     0.00228    0.00004    0.12833

                     8      1      55      7     0.00032    0.00000    0.07718

                     9      1      55      8     0.00005    0.00000    0.05836

                    10      1      55      9     0.00001    0.00000    0.04725

                    11      1      55     10     0.00000    0.00000    0.03765

                    12      1      55     11     0.00000    0.00000    0.02576

                    13      1      55     12     0.00000    0.00000    0.02029

                    14      1      55     13     0.00000    0.00000    0.01983

                    15      1      55     14     0.00000    0.00000    0.02057

                    16      1      55     15     0.00000    0.00000    0.01759

                    17      1      55     22     0.00000    0.00000    0.01846

                    18      1      55     30     0.00000    0.00000    0.02035

                    19      1      55     31     0.00000    0.00000    0.02128

                    20      1      55     32     0.00000    0.00000    0.02209

                    21      1      55     34     0.00000    0.00000    0.02205

                    22      1      55     35     0.00000    0.00000    0.02344

                    23      1      55     36     0.00000    0.00000    0.01616

                    24      1      55     43     0.00000    0.00000    0.04070

                    25      1      55     53     0.00000    0.00000    0.06515

                    26      1      55     54     0.00000    0.00000    0.26885

                    27      1      55     57     0.00000    0.00000    0.13609

                    28      1      55     58     0.00000    0.00000    1.00000

                    29      0      22      0     1.00000     .          .

                    30      0      22      1     0.98117    0.96556    0.99703

                    31      0      22      2     0.97348    0.95206    0.99537

                    32      0      22      3     0.95402    0.91989    0.98941

                    33      0      22      4     0.94392    0.90325    0.98642

                    34      0      22      5     0.92202    0.86928    0.97795

                    35      0      22      6     0.91467    0.85776    0.97535

                    36      0      22      7     0.88863    0.81923    0.96391



………

13. Graph these survival curves.

proc gplot data=outdata3;

title 'Survivor f3unction at age 55 with drug';

plot S*time LCL*time UCL*time/overlay vaxis=axis1;

symbol1 i=join c=black line=1;

symbol2 i=join c=black line=2;

where age=55;

run;

proc gplot data=outdata3;

title 'Survivor function at age 22 without drug';

plot S*time LCL*time UCL*time/overlay vaxis=axis1;

symbol1 i=join c=black line=1;

symbol2 i=join c=black line=2;

where age=22;

run;

Now, let’s switch datasets to UIS.  The data dictionary is below for your reference:

These data were collected from a randomized trial of two in-patient treatment courses for drug addiction (heroin, cocaine, and unspecified other drugs): a shorter treatment plan and a longer treatment plan.  Time to censoring or return to drug use was recorded, as was length of time in the treatment plan.  Variables are below:

	Variable
	Description
	Codes/Values

	ID
	Identification code
	1-628

	AGE
	Age at enrollment
	Years

	BECKTOTA
	Beck Depression Score at Admission
	0-54

	HERCOC
	Heroin/Cocaine use during 3 months prior to admission
	1=Heroin& cocaine

2=Heroin only

3=Cocaine Only

4=neither



	IVHX
	IV drug use history at admission
	1=never

2=previous

3=recent

	NDRUGTX
	Number of prior drug treatments
	0-40

	RACE
	Subject’s race
	0=White

1=Other

	TREAT
	Treatment randomization Assignment
	0=short

1=long

	SITE
	Treatment site
	0=A

1=B

	LOT
	Length of treatment

(exit date-admission date)
	Days



	TIME
	Time to return of drug use

(Measured from admission)
	Days

	CENSOR
	Returned to drug use
	1=Returned to drug use

0=Otherwise


14. Examine data using a KM plot stratified by “TREAT” which specifies the randomization group (short or long).

goptions reset=all;

proc format;

value treat

1 = "long"
0 = "short"; 

run;

proc lifetest data=hrp262.uis plots=(s) graphics censoredsymbol=none;


label time='time(days)';


time time*censor(0);


strata treat; 


title 'UIS KM plot';


symbol1 c=black v=none line=1 i=join; 

symbol2 c=black v=none line=2 i=join; *gives plots that will be suitable for black and white copying;



format treat treat.;

run;

15. Run Proc PHREG with treat and age as predictors, and then add site as a stratification variable.  Scroll through and review output as a class.

proc phreg data=hrp262.uis;

model time*censor(0)=age treat/risklimits;

run;
proc phreg data=hrp262.uis;

model time*censor(0)=age treat/risklimits;

strata site;

run;


WITHOUT STRATIFICATION BY SITE

                            Analysis of Maximum Likelihood Estimates

                  Parameter    Standard                            Hazard   95% Hazard Ratio

   Variable  DF    Estimate       Error  Chi-Square  Pr > ChiSq     Ratio   Confidence Limits

   age        1    -0.01327     0.00721      3.3847      0.0658     0.987     0.973     1.001

   treat      1    -0.22298     0.08933      6.2307      0.0126     0.800     0.672     0.953

WITH STRATIFICATION BY SITE
                            Analysis of Maximum Likelihood Estimates

                  Parameter    Standard                            Hazard   95% Hazard Ratio

   Variable  DF    Estimate       Error  Chi-Square  Pr > ChiSq     Ratio   Confidence Limits

   age        1    -0.01429     0.00729      3.8434      0.0499     0.986     0.972     1.000

   treat      1    -0.23507     0.08961      6.8820      0.0087     0.791     0.663     0.942

16. Run PROC PHREG with treatment as a time-dependent variable and age. 

proc phreg data=hrp262.uis;

model time*censor(0)=off_trt age treat/risklimits;

if lot>=time then off_trt=0; else if lot<time then off_trt=1;

run;


RESULTS:

                            Analysis of Maximum Likelihood Estimates

                  Parameter    Standard                            Hazard   95% Hazard Ratio

   Variable  DF    Estimate       Error  Chi-Square  Pr > ChiSq     Ratio   Confidence Limits

   off_trt    1     2.56681     0.15173    286.1826      <.0001    13.024     9.674    17.535

   age        1    -0.00719     0.00728      0.9777      0.3228     0.993     0.979     1.007

   treat      1     0.01082     0.08988      0.0145      0.9042     1.011     0.848     1.206



Defines survivor curve for man age 55 with past IV drug use. Defines it at all 28 observed event times.





logS(t=2)





log(-logS(t=15))





9.6% increased hazard for each 1-year of age; 156.3% increase in hazard if used drug.





Wald Tests: tests of whether or not individual betas are equal to 0.





� EMBED Equation.3  ���





� EMBED Equation.3  ���





risklimits option asks for confidence limits for the estimated hazard ratio.





Note use of proc format to assign format to the variable treat.





Review of output from PHREG





Recall: in SAS LAB 2 we graphed log(-log)S(t) for these data against log(time) using proc lifetest





Interpretation: there is a 20% decrease in the rate of relapse in those treated in the long program compared with those in the short program.





It’s a quick drop-off for this group.  Note that the confidence intervals go haywire around PredictedSurvival=0.





Note use of the very convenient where statement (works in most SAS procs) for subsetting.





log(-logS(t=2))





Syntax for PHREG





Note that DRUG has been removed from the model statement; stratifying by DRUG allows SAS to assume different baseline hazards for each drug group (which can later be compared to test PH assumption)…





Short because no events in this group beyond t=20





Mean value of age for the dataset





As in PROC LIFETEST, you can categorize continuous variables within the Strata statement.  Here, asks SAS to divide into age groups: [-� EMBED Equation.3  ���,30) [30,40) [40,� EMBED Equation.3  ���)





p-value is not significant, so there is little evidence of an interaction exists.  





Function curves upward, indicating an increasing hazard with time.





� EMBED Equation.3  ���





Mean value of drug for the dataset





28 unique event times





-2 Log Likelihood for comparing with other models (likelihood ratio test)





Test of the global null hypothesis that all coefficients are equal to 0.





Method used for dealing with ties; Breslow is default





9.6% increase in hazard for every 1-year increase in age, controlling for the effect of IV drug use; 156.3% increase in hazard for using IV drugs, controlling for age.





Note: only small differences between Breslow, exact, and Efron; Breslow and Efron approximations attenuate slightly toward the null.


Discrete gives odds ratios; hence the small increase in size.





Stratification allows there to be totally different (and even non-parallel) baseline hazard functions in each stratum. Rather than trying to estimate hazard ratios between the strata, you instead generate a hazard ratio that has been “averaged” over the different strata (like a mantel-haenszel odds ratio).


Risk sets for the partial likelihood are allowed to be stratum-specific.





This is useful if you want to control for a possible confounding variable that (1) may violate the Proportional Hazards assumption and/or (2) is a nuisance variables—that is you don’t care about getting hazard ratios for it.





For example, in this study, there were two hospitals in which treatment was administered (site 1 and site 2).  There’s good reason to believe that baseline hazards might be totally different depending on the site (because of different local populations, and different ways of administering the treatment).  For this study, we don’t care about the effect of site; we just want to know if the long treatment is better than the short treatment. So, we “average” the HR for the long treatment compared with the short treatment over the two different sites.





SAS automatically creates time-dependent variables for you in the phreg procedure (without a separate datastep). This temporary variable is not saved.





**Use Solutions(Analysis(Interactive Data Analysis(Work.Outdata


to examine the contents of outdata. What is stratification doing?





This is adding a time-dependent covariate: off_trt, which is 1 if they have left the treatment program and 0 if they are still in the treatment program.  This is evaluating whether just being in the treatment facility and program prevents a relapse.





At each event time T, for each individual still in the risk set, SAS compares the length of treatment to T.  For example, the first event occurred at 2 months. SAS goes through and compares the length of treatment for each individual with 2 months. If the person was in treatment longer than 2 months, then they are still on treatment at the first event time and off_trt=0. If the person was in treatment shorter than 2 months, then at 2 months they are off treatment and off_trt=1. Thus, the values of off_trt change at each event time and in each likelihood term of the likelihood equation.








The fact that there’s little difference between the hazard ratios with or without stratification indicates that there’s little confounding by site, BUT it might be important to leave the stratification in anyway, to avert potential criticism by reviewers of your analysis.





There appears to be no difference between short vs. long treatment, beyond the fact that currently being in treatment prevents relapse and the longer program keeps them in longer.








There seems to be a protective effect for being randomized to the longer treatment group.

